FORMAT

A Magazine from INDUG
For DISCiPLE & PLUS D Users

ISSUE #7 - FEBRUARY 1988

THE MAG WITH MORE BYTE...

INDUG.

FORMAT

A Magazine from INDUG
For DISCiPLE & PLUS D Users

ISSUE #5 - DECEMBER 1987

Bumper Christmas Issue

INDUG.

FORMAT

A Magazine from INDUG
For DISCiPLE & PLUS D Users

ISSUE #9 - APRIL 1988



TAPE TO DISC
WE SHOW YOU HOW

INDUG.

FORMAT

A Magazine from INDUG
For DISCiPLE & PLUS D Users

ISSUE #8 - MARCH 1988



A TOUCH OF ART...

INDUG.

FORMAT

A Magazine from INDUG
For DISCiPLE & PLUS D Users

INTRODUCTORY ISSUE



Helps You Get The Best From
Your Interface

INDUG.

CONTENTS

INTRODUCTORY ISSUE

FORMAT is published by INDUG. 34 Bourton Road, Gloucester, GL4 OLE, England. Telephone 0452-412572. **DISCiPLE** and **PLUS D** are trade marks of MILES GORDON TECHNOLOGY, Lake Side Technology Park, Phoenix Way, Swansea, South Wales, SA7 9EH. The DISCiPLE interface is marketed by Rockfort Products, 81 Church Road, Hendon, London, NW4 4DP.



A very warm welcome to this the introductory issue of FORMAT, the magazine of INDUG, the Independent DISCiPLE and PLUS D Users Group.

INDUG was formed in May 1987, first to support users of the DISCiPLE and, more recently, the MGT PLUS D. As this is written (in June 1988) we have around 1000 members world wide and we are still growing fast. Our aim is to help you get the most from your Spectrum computer, your DISCiPLE or PLUS D, and the GDOS disc operating system.

FORMAT is published monthly and is free to club members. Although each issue will concentrate on the DISCiPLE/PLUS D and the GDOS operating system, we also carry lots of other items of a more general Spectrum nature. There are feature articles, hints & tips, software reviews (not games), special offers and competitions. We will try to answer your problems either through the regular HELP page or by our telephone 'HOT LINE' service, full details of which will be sent to new members.

Since its launch, the DISCiPLE has done more than any other add-on before to expand the horizons of Spectrum users. The PLUS D has extended the benefits of a powerful disc system to an even wider audience. Many software companies have now converted their programs to full GDOS compatibility and others are doing so all the time. INDUG helps software companies in this conversion by providing technical help and support.

We would like to stress that while working very closely with Miles Gordon Technology, INDUG is completely independent and exists to serve the interests of its members.

In the following pages we try to give you a glimpse of the sort of articles published in FORMAT. However, please remember that space is very limited in this introductory issue so it is impossible to do justice to the wide range of articles that appear in regular issues FORMAT. From beginners to machine code addicts, there's something for everyone in FORMAT. Where an article refers to the DISCiPLE it, of course, applies to the PLUS D as well.

I hope you enjoy this issue and we look forward to welcoming you as a club member. A membership form is included in this issue or you can write to us at the club address on page 2. All back issues are available and details will be sent with your first issue.

Bob Brenchley. Editor.

DISC or DISK?
That is the Question.

Which is correct DISC or DISK? Is there in fact a real answer? Do you really care?

In the early days of computing a DISC was a large solid unit that could be screwed into an even larger and more solid drive. They consumed hundreds of watts of power, took some time to 'spin-up' to the right speed and cost more than any of us earn in a year or two (or in my case, ten).

Then in the USA the old 'BIG BLUE' IBM invented the 8 inch DISKETTE or FLOPPY DISK as it came to be known. This later shrunk to 5.25 inch which is now the world standard, although 3.5 inch are catching up fast. In the US the correct term is DISK but they still refer to a DISC DRIVE, you see the DISC is in the DISKETTE and the DRIVE drives the DISC not the KETTE, clear now? good I'm glad you understood that bit.

In the UK our computer giant, ICL, always used DISC for all versions. The BBC, or should I say Acorn, did a cop-out and allowed both the DISC and DISK command to be used, but still referred to DISC in most of their manuals.

So what is MY answer?, well as was trained (through ICL) to spell it DISC and I am true blue British.. I will continue to use DISC but you can make your own mind up, after all we all know DISC or DISK is still better than TAPE.

Ed.

* * * * * * * * * * *

BARGAIN CORNER

YOUR ADVERTS YOUR ADVERTS YOUR ADVERTS YOUR ADVERTS YOUR ADVERTS
YOUR ADVERTS YOUR ADVERTS YOUR ADVERTS YOUR ADVERTS YOUR ADVERT

Buying, Selling, Pen Friends, Clubs, etc.

This space is reserved for you. Any PRIVATE advert of up to 30 words (subject to acceptance) will be run in the next available issue of FORMAT free of charge. Any software sold must be original copies, with full instructions and in working order. The publishers will not be held, in any way, responsible for adverts in this column.

POKE @6,n
OR - HOW TO UNLOCK YOUR PRINTER

Because a BASIC program can contain embedded colour control characters, and other unprintable rubbish, the GDOS printer channel has to ignore ASCII codes less than 32 (see appendix A of the Spectrum 48k Basic Programming Manual) when LLISTing or LPRINTing. Only code 13, the carriage return code, is sent unaltered, the rest will normally be used by the system (TAB, AT etc) or screened out, some however will print a question mark.

Now thats fine in most cases but what do you do when you want to output Printer Control Codes. Most printers have different print modes or styles and the time will come when you will want to use them. Well if there were only a few codes, you could use the method given in the manual and output CHR$(27) if front of each code. I.E. to output codes 27,45,1 (underline ON for an Epson printer) you would enter:-

```
LPRINT CHR$(27);CHR$(27);CHR$(27);CHR$(45);CHR$(27);CHR$(1)
```

(Note that even CHR$(27) needs a CHR$(27) before it to work)

OK so far but what if you are using a wordprocessor like TASWORD 2 which only allows four numbers per printer function? or perhaps the codes are already built in to a commercial program. Well, this is where the GDOS command POKE @6,n comes in. POKE (on the 'O' key) followed by @ (symbol shift 2) is used to control certain GDOS variables in an area of shadow RAM. Location 6 in this area tells the printer routines to ignore codes less than 32 unless proceeded by a code of 27.

The default value of location 6 is zero but if we change its value by doing a POKE @6,1 then all codes will be output to the printer without interference by the PLUS D. So to output an underlined message to an Epson printer enter the following:-

```
POKE @6,1:LPRINT CHR$(27);CHR$(45);CHR$(1);"I LOVE MY PLUS D
";CHR$(27);CHR$(45);CHR$(0): POKE @6,0
```

Remember POKE @6,0 when you have finished printing just in case you need to do an LLIST later on.

POKE @6 could be entered as part of a BASIC program or as a direct command prior to loading a machine code program. You could try making it part of your 'AUTOLOAD' program for any disc that needs it. With Tasword 2 insert POKE @6,1 as a new line 279 and POKE @6,0 as line 281, your printer control codes will now get through to your printer without interference.

HAPPY PRINTING

B.B.

THE HELP PAGE

Problems with your DISCiPLE, PLUS D or Spectrum. Dont worry, let the HELP PAGE sort them out. Note: One question per letter please.

SOME COMMON PROBLEMS.

On the whole the DISCiPLE and PLUS D are very reliable pieces of hardware, but there are some small problems which may crop up with some users.

On this page each month we try to cover some of the areas where users seem to encounter difficulties. Problems answered relate to the PLUS D / DISCiPLE and the SPECTRUM in general.

The most common problems relate to the dreaded "SECTOR error" message. This will usually appear when loading a file from disc or when using the SAVE command to 'COPY' a file. First make sure the disc is one that is matched to your drive, check the density and number of sides are correct. Most single sided discs will format as double sided, BUT its false economy if the only copy of your program is on a cheap disc and wont load. The same comments apply to single/double density discs. If you buy Double Sided, Double Density, 96 Tracks Per Inch discs (DS/DD 96TPI) you can't go wrong.

If the disc matches the drive then check the stepping rate you set up is correct. Try 12ms to start with and then 24,36 or even 48ms (you could go as slow as 255ms but all drives we have been able to try work at 48ms or faster). If the problem persists even when using a new disc try putting the drive as far away from the T.V. as you can, and make sure no power cables cross over the ribbon cable connecting it to the interface. Still got a problem? well get a Head Cleaning Disc (from Boots or most other computer stores) use it as directed and if this cures the problem remember to use it at regular intervals thereafter.

If your drive is one of the old style (full height) 5.25" units it may not format in the double density mode the PLUS D uses, try to borrow another drive and see if this make a difference.

Lets now turn to printers. These useful things sometimes produce more problems than all other peripherals put together. There is no accepted standard in printer control codes, some printers are called EPSON Compatible but what EPSON are they compatible with? MX80; RX80; FX80 or what? If you follow, step by step, the system program which comes with your interface you should be able to set up your DOS for most printers. The most common fault in setting up is to enter a control code wrong, don't worry you wont damage your printer. Simply re-run the system program and try again.

If your printer has Bit Image Graphics then make sure you select the right density, you need the SINGLE density mode which on most printers gives 480 bits per line.

P.S. There are currently lots of cheap 3 inch drives floating around. Some of these were produced for the Amstrad CPC range of computers and will not work with your interface. Check before buying, that any 3 inch drive is guaranteed to work with the DISCiPLE or PLUS D, better still try one before parting with your money.



Contributions from FORMAT readers are very welcome. We like to publish articles on any subject relating to the DISCiPLE, the Spectrum or indeed any aspect of computing that you feel may be of interest to other FORMAT readers.

Some points to bear in mind

* Ideally submit your article as a Tasword (2 or 3), The Last Word, or similar text file (with a printed copy). We can accept disc (5.25 or 3.5) or tape.
* Any graphics or diagrams should be drawn in black at twice (2x) normal size for reduction. Pack well, Do Not crease.
* Mark everything with your name, address and telephone number.
* Keep a copy, DO NOT send your only version, the post office is not that good (and nor are we).
* Include a stamped addressed envelope if you want your material returned.
* Remember to say which version of G+DOS your program/article was written for and if 48k or 128k Spectrum.
* Feel free to contact us with your ideas before committing yourself to writing that long article.
* DO NOT COPY ITEMS FROM OTHER MAGAZINES.

Remember we pay for all articles and programs published in FORMAT, so get writing.

PROGRAM. PAGE. . PROGRAM. PAGE. . PROGRAM . PAGE. . PROGRAM. PAGE. . PROGRAM. PAGE. .

The following program is a very useful disc menu utility. It works by holding up to 20 Basic program names in an array (F$) and loading the file selected by a single key press. You can add or delete an entry without affecting what is on the disc, you can catalogue the disc and of course resave the program either as a file called 'MENU' or as 'autoload' so all you need to do is switch on, insert your disc, RUN, and the menu will be displayed. If you use this on a disc full of SNAPSHOT file then add an S or K to the LOAD statement in line 190




```
  1 REM MENU V1.4 1(C)1987 INDUG.
  3 REM TO RESTART USE 'GOTO 30'
  4 REM FOR ALL VERSIONS OF GDOS.
 10 DEF FN A(A)=A-10*(A>10)
 11 DEF FN B(A)=A>10
 20 DIM F$(21,10)
 30 POKE 23658,8: BORDER 1:
 40 PRINT TAB 5; PAPER 6; INK 1; BRIGHT 1;"DISCiPLE PROGRAM LOA
DER"
 50 PRINT '"PRESS   1 - To enter new title."
 60 PRINT "         2 - To delete title."
 70 PRINT "         9 - To CATalogue Disc."
 80 PRINT "         0 - To resave loader."
 90 PRINT '"  OR   Letter to load program."'''
100 LET N=0: FOR I=1 TO 20: IF F$(I)="[10SPACES]" THEN GOTO 120
110 PRINT AT 9+FN A(I),2+FN B(I)*16;CHR$ (64+I);":- ";F$(I): NE
XT I
120 LET N=I: PRINT #0;TAB 7; INVERSE 1;"   PRESS KEY NOW   "
130 GO SUB 290: IF I$="0" THEN PRINT AT 5,8; INVERSE 1;I$: GOTO
320
140 IF I$="1" THEN PRINT AT 2,8; INVERSE 1;I$: GOTO 200
150 IF I$="2" THEN PRINT AT 3,8; INVERSE 1;I$: GOTO 240
160 IF I$="9" THEN CLS : CAT *!: PAUSE 0: GOTO 30
165 IF I$="8" THEN INPUT "NEW DRIVE NUMBER = ";D: CLS : CAT D!:
PAUSE 0: GOTO 30
170 IF I$<"A" OR I$>CHR$ (63+N) THEN GOTO 130
180 LET I=CODE (I$)-64: PRINT AT 9+FN A(I),2+FN B(I)*16; INVERS
E 1;I$: GOTO 190
190 LET X=VAL "CODE (I$)-64": POKE 23658,0: LOAD D*;F$(X): REM
LOAD PROGRAM
200 IF N=21 THEN PRINT #0;AT 0,8; FLASH 1;"NO MORE ROOM": PAUSE
200: GOTO 30
210 INPUT "Program Name:- ";I$: IF I$="" OR LEN (I$)>10 THEN GO
TO 200
220 IF I$=" STOP " THEN GOTO 30
230 LET F$(N)=I$: GOTO 30
240 IF N=1 THEN GOTO 30
250 INPUT ;: PRINT #0;"PRESS LETTER TO DELETE"
260 GO SUB 290: IF I$=" STOP " THEN GOTO 30
270 IF I$<"A" OR I$>CHR$ (63+N) THEN GOTO 260
280 LET K=CODE I$-64: FOR I=K TO 20: LET F$(I)=F$(I+1): NEXT I:
GOTO 30
290 LET I$=INKEY$: IF I$<>"" THEN GOTO 290
300 LET I$=INKEY$: IF I$="" THEN GOTO 300
310 RETURN
320 PRINT AT 21,1; PAPER 2; BRIGHT 1;" SAVE AS 'AUTOLOAD' "; FL
ASH 1;"Y"; FLASH 0;"ES OR "; FLASH 1;"N"; FLASH 0;"O "
330 IF INKEY$<>"" THEN GOTO 330
340 LET R$=INKEY$: IF R$="" THEN GOTO 340
350 IF R$="Y" THEN SAVE d*"autoload" LINE 30: VERIFY d*"autoloa
d": GOTO 30
360 IF R$<>"N" THEN GOTO 330
370 SAVE d*"MENU" LINE 30: VERIFY d*"MENU": GOTO 30
```

Playing with the DISCiPLE's screen copy snapshot, trying to print a poster for a local disco, I needed a double height print routine. The following is my answer, it works by creating new character sets in 1536 bytes starting at 63232. One set holds the top half, the other the lower half. T$ holds the text to be printed, TX & TY hold the line and column where text is to be printed. Printing is then done by using the first set to print the top half of all the characters on a line and then switching sets and printing the bottom half. Remember to reset to the ROM character set after printing double height or you will have difficulty reading anything printed at single height.


```
  1 REM DOUBLE HIGHT CHARACTERS
  2 (C)1987 RON MARKS.
 10 GOSUB 1000
 20 CLS
 30 LET T$="DISCiPLE": LET TX=3
 40 GOSUB 2000
 50 STOP
1000 REM SET UP CHARACTER SETS
1010 FOR I=0 TO 767 STEP 8
1020 FOR N=0 TO 7
1030 POKE (63232+I+N),PEEK (15616+I+INT (N/2))
1040 POKE (64000+I+N),PEEK (15616+I+INT (N/2)+4)
1050 NEXT N: NEXT I: RETURN
2000 REM PRINT TEXT
2010 LET TY=INT ((32-LEN T$)/2)
2020 FOR K=0 TO 1
2030 POKE 23607,246+3*K
2040 PRINT AT TX+K,TY;T$
2050 POKE 23607,60: NEXT K: RETURN
```



HINTS & TIPS

Readers Hints & Tips to help everyone get the most from their computing.

Send your Hint or Tip for the DISCiPLE, PLUS D or Spectrum to the address on page 2. Try to keep them short and sweet so we can get as many in as possible.

NAME THAT DISC

As the DISCiPLE has no direct facility to name a disc it is sometimes difficult to know which disc you are using without removing it from the drive to read the label. The answer to this problem came when I found the ROM looks for a system file using the form 'SYS*' (*=Wild-card). this means the last seven characters of the system file name can be anything you like. I number my discs by FORMATing a disc then doing a: SAVE d1"SYS £nn" CODE 0,6656 where nn is my disc number, now when ever I do a disc CATalogue I can identify the disc straight away.

Steve Martin. Cornwall.

Nice one Steve, PLUS D users may like to know that they get six characters spare, as the BOOT routine looks for "+SYS*". Autoload files can also be used as only "AUTO*" is looked for.

BOOTing

When, having loaded a program from tape ready for transfer to disc, you find you have forgotten to BOOT the disc operating system, entering RUN then merely runs the program. However you don't have to switch off and start again. The answer is to enter RUN with a high invalid (over 9999) line number. So typing "RUN 10000" and then ENTER will BOOT up your system, without losing the program in memory provided there was no AUTOload file on the disc. If there is an AUTO file then "RUN boot" will do the job on version 3 DISCiPLEs and the PLUS D.

HAROLD BURTON. Edinburgh.

ONE LINER

This one line program gives a full disc verify for a double sided, 80 track drive. I use it to test a disc before saving valuable data, any dud sectors will produce an error.

```
10 FOR i=0 TO 128 STEP 128: FOR t=0 TO 79: FOR s=1 TO 10: LOAD @
1,i+t,s,50000: NEXT s: NEXT t: NEXT i
```

K.R. TYNE. LONDON.

Spread a little happiness among other users. Send in your HINTS and TIPS for the DISCiPLE, PLUS D or SPECTRUM. Keep them short and to the point so we can get as many in as possible.

This Page depends on YOUR contributions.

AN OPEN & CLOSE CASE

The statement OPEN#4;d1"test" will open a file called "test" on drive 1. If the file already exists it will be opened as an input file, if the file does not exist then an output file is created. On microdrives if you already had a file of the same name on the cartridge you had to ERASE it before OPENing it if you wanted to write a file, a messy way of doing things. The DISCiPLE/PLUS D adds two extensions to the OPEN# syntax, IN and OUT (using the BASIC keywords). OPEN#4;"test" OUT will force the file to be opened as a write file and if it already exists will prompt you with the OVERWRITE Y/N question which by the way now prints the filename as part of the prompt. OPEN#4;d1"test" IN opens a read file, if the file is not found an error message is given. You can have up to 12 files OPEN at once (Streams #4 to #15) and, like Interface 1, each file has a buffer in the Channels Area so the more files you have the more memory they take up.

Due to the way that GDOS keeps track of free sectors when writing files to disc you can only use one drive at a time for OUT files although you can have more than one OUT file on the same drive. There is no such restriction on IN files as GDOS does not need to keep a record of free space for these.

Note that unlike microdrives you can't redirect stream #3 (the printer stream) to an output file.

Having OPENed a file you need to be able to do something with it of course. PRINT#n will write data to Stream n if the file is an OUT file, INPUT#n or INKEY$#n will read data from an IN file.

Once you have finished with a file you will need to CLOSE it. CLOSE#*4 will close the file OPENed on stream #4, if it was a write file the current buffer is written to disc, the disc directory is updated with the files details and the Channel Area allocated is recovered. Once a stream is CLOSEd its number can be used again if required. CLOSE#* without a stream number will close all current files, any data in the buffers for output files will be written to the disc. To prevent corrupt (unclosed) output files in GDOS, the RUN command does not clear the Channels. You must CLOSE files to free the Stream number.

When an OUTput file is used the directory entry is not made until the CLOSE# command is issued. It is therefore good practice to do a CLOSE#* if your program crashes to preserve any data you have already written.

OK so lets see this in action, type in the following short program and try it out.

```
10 OPEN#6;d1"testfile" OUT
20 FOR I=1 TO 10
30 INPUT "Type in a number please ";NUM
40 PRINT#6;NUM
50 NEXT I
60 CLOSE#*6
```

```
 70 OPEN#6;d1"testfile" IN
 80 FOR I=1 TO 10
 90 INPUT#6;data
100 PRINT "For input ";I;" you entered ";data
110 NEXT I
120 CLOSE#*6
```

Line 10 opens an OUT file. Lines 20 to 50 ask you for a series of numbers and then writes them to the file. Line 60 closes the file and frees the stream. The rest of the program opens the same file as an IN file, reads the number you entered from disc and prints it to the screen. A bit boring I know but it does demonstrate the principle.

Files created by an OPEN# command appear in the full directory list as 'OPENTYP' and with up to 780k of disc space you can now store an awful lot of data.

The microdrive syntax OPEN#n;"M";drive;"FILENAME" will work, although you loose the benefit of the IN or OUT extensions, but CLOSE# must have the * inserted to fail the Spectrum ROM syntax which has a fatal bug in it.

Right thats the simple stuff over with now lets deal with some more advanced matters. It would be useful not to read beyond the last item of data on an IN file. This would avoid the 'End of File' message which would stop your program.

The secret to avoiding this problem hides away in the channel area created when the file is opened. Three bytes, originally stored in the directory when the file was written, hold the High, Middle and Low bytes of a count of the number of characters in the file. The following subroutine returns the number of characters left on the file, if this is zero then any further attempt to read from that file will give the EOF error.

```
2000 REM enter with STN=stream number.
2001 REM exit with CL = characters left
2010 LET OFFSET=PEEK (23574+STN*2)+256*PEEK (23575+STN*2)
2020 LET CHANADR=PEEK (23631)+256*PEEK (23632)+OFFSET-1
2030 IF CHR$(PEEK (CHANADR+4))<>"D" THEN PRINT"Not a disc
file":STOP
2040 LET CL=PEEK (CHANADR+31)+256*PEEK (CHANADR+32)+65536*
PEEK (CHANADR+18)
2050 RETURN
```

Now add this line to the test program given earlier and RUN it.

```
105 LET STN=6:GOSUB 2000:PRINT CL;" Chrs left in file."
```

Line 2030 tests that stream #6 is attached to a disc file Notice that the count of characters include the carriage return at the end of each field.

If you need to keep a check on the number of characters written to an OUT file the count is stored in CHANADR + 231 (low), 232 (middle) and 219 (high).

The routine could be adapted to cope with both IN and OUT files. Location 9 + 256 * Location 10 will equal 551 if the file is an input or 787 if it is an output file.

I hope this article has helped you to understand how OPEN type files are used. I think we will be returning to them many times in the future.

WILL IT WORK?

When any new computer comes onto the market it will usually succeed (or fail) on the strength of its software base. NO SOFTWARE = SMALL COMPUTER SALES. The same must be true when it comes to marketing a disc interface, if little software will work then few people will feel it worth investing in one. This has been true of many Spectrum disc interfaces in the past, most failed because little software would work with them.

First the good news. By making the PLUS D software compatible with the DISCiPLE, which in turn was as compatible as possible with Microdrives, most software will work with little or no alteration. In fact any MICRODRIVE program that uses Interface One 'HOOK CODES', or one using BASIC to load and save files, will work without any problems. The difficulties start when a programmer decides to use a direct call to the Shadow ROM on Interface 1, most professional programmers would never attempt to use a ROM routine (you can't count on the ROM staying unaltered) but some throw caution to the wind and use 'ILLEGAL' calls. Now the Interface 1 ROM changed several times but still some programs exist that break the rules. A program such as TASWORD 3 will not work because of calls to CAT the Microdrive.

Now for the even better news. Because the DISCiPLE has been on sale for over a year, many software companies have converted there programs to use the full facilities of GDOS and even more are in the process of doing so. The 'WILL IT WORK' section will list programs that have been tried and tested either by us or by our readers. If you try any program and find it works why not drop us a line with details of the program, and any alterations you made, so we can add it to the list.

In the following list some programs require modification before they will work, but in most cases this only involves changing a few lines of basic, some have been the subject of articles in FORMAT. Programs marked with a '*' are available in special DISCiPLE / PLUS D versions.

PROGRAMS TESTED AND WORKING.

TASWORD 2 (Tasman Software.); OMNICALC 2* (Microsphere.); QUALITAS (Seven Stars Publishing); POWERPRINT II (Buttercraft Software); DISC MANAGER* (Better Bytes); DEVPAC*, PASCAL*, 'C'* (Hisoft); THE LAST WORD* (Trojan/Myrmidon); PLUS 80 EDITOR / ASSEMBLER*, ADDRESS MANAGER*, FINANCE MANAGER*, WORD MANAGER* (OCP now ECC); MASTERFILE, DLAN (Campbell Systems); SMALL BUSINESS ACCOUNTS (Sinclair); DISK_FILE* (Redcliffe Software); LETTA HEAD PLUS, DUMPY, LINE-O-TYPE (Bradway Software); VUFILE, VUCALC, VU-3D (Psion) + MANY MANY MORE...

In addition we also have a growing range of software specially produced for INDUG members. Full details will be sent with your first issue of FORMAT.